*** Make flat line projection estimate

cd /disk/agedisk3/medicare.work/poterba-DUA52260/jetson-dua52260/botox/synth/100pct

use growthgroupspanel.dta, clear
keep if group == 999 


local treatment_period = mofd(mdy(6,5,2007))
disp `treatment_period'


** Create flat line projection
gen last12 = pmt_amt if `treatment_period' - dt <= 12 & `treatment_period' - dt > 0 
egen flatline = mean(last12)
replace flatline = . if dt < `treatment_period' - 12

** Produce difference in series
gen deterrence = flatline - pmt_amt if dt >= `treatment_period'

** Time discount the difference at 10% 
local r = 1.1^(1/12)

gen deterrence_disc = deterrence*1/(`r'^(dt-`treatment_period'))
replace deterrence_disc = . if dt-`treatment_period' >= 60
count if !mi(deterrence_disc)
assert(r(N)==60)

*** Compute total
log using robustness/flatline/deterrence_flatline.log, replace
preserve
collapse (sum) deterrence_disc
qui summarize deterrence_disc
local deterrence = r(min)
disp `deterrence'
restore
log close

*** Truncate graph at 5 years 
drop if dt  - `treatment_period' >= 60
summarize dt 
local minperiod = r(min)
local maxperiod = r(max)

*** Settlement: 8/31/10
local settlement = mofd(mdy(8,31,2010))
format dt %tmMon_YY


twoway (line pmt_amt dt, lcolor(navy) lwidth(thick)) (line flatline dt, lwidth(thick) lcolor(maroon) lpattern(dash)) (rarea pmt_amt flatline dt if dt>= `treatment_period', color(gs5)), xline(`treatment_period') xline(`settlement', lcolor(black)) legend(order(1 "Botox" 2 "Flat Line Projection") width(80)) xscale(range(`minperiod' `maxperiod')) xlabel(504 528 552 `treatment_period' 588 612) xtitle("Month") ytitle("$/Month") graphregion(color(white)) yscale(range(0 7000000)) ylabel(0 3000000 6000000, format(%13.0fc))



graph save "robustness/flatline/BotoxOverlay_FlatLine.gph", replace

